234

252


17

PROGRAMARE MIXTĂ: C – LIMBAJ DE ASAMBLARE

17.1 Programul principal este scris în limbaj de asamblare, procedurile apelate sunt scrise în limbajul C


Variabilele transmise procedurii apelate se definesc cu atributul extrn. Procedurile apelate se definesc, de asemenea, cu atributul extrn. Se folosesc pentru apel proceduri scrise în limbajul C pentru a realiza funcţii de intrare/ieşire care presupun conversii.

Programul ASM1 adună trei numere. Se apelează funcţia scanf() pentru a iniţializa trei variabile a, b, c. După însumare se apelează procedura printf() pentru a imprima variabila suma. Se definesc în programul C variabilele de tip char: format1, format2, format3, format4.

;ASM 1

.model small

.stack 100h

.data

; date externe

  extrn c format1:word

  extrn c format2:word

  extrn c format3:word

  extrn c format4:word

  extrn c format:word

  extrn c a:word

  extrn c b:word

  extrn c c:word

;date locale

  suma dw ?

.code

  public _main

; functiile externe apelate

  extrn c printf:proc

  extrn c scanf:proc

  extrn c sablon:proc

  extrn c refac:proc

; intrarea in program

  _main proc

; apel functie C de formatarea a ecranului

    call sablon c

; citirea parametrilor de la tastatura

    lea ax,format1

    call printf c,ax

    lea ax,format

    lea bx,a

    call scanf c,ax,bx

    lea ax,format2

    call printf c,ax

    lea ax,format

    lea bx,b

    call scanf c,ax,bx

    lea ax,format3

    call printf c,ax

    lea ax,format

    lea bx,c

    call scanf c,ax,bx

; insumarea celor trei valori

    mov ax,a

    add ax,b

    add ax,c

    mov suma,ax

; afisarea rezultatului

    lea ax,format4

    call printf c,ax,suma

; apel functie C de refacere a ecranului

    call refac c

; iesirea din program

    ret

  _main endp

end

//C1

#include <stdio.h>

#include <conio.h>

char format1[20]="a=";

char format2[20]="b=";

char format3[20]="c=";

char format4[20]="\nsuma=%d\n";

char format[20]="%d";

int a,b,c;

void sablon(void)

{

  int i;

  clrscr();

  for(i=0;i<80;i++)

    printf("-");

  printf("Introduceti datele:\n");

  for(i=0;i<80;i++)

    printf("-");

  printf("\n\n\n");

  for(i=0;i<80;i++)

    printf("-");

  printf("\n");

  for(i=0;i<80;i++)

    printf("-");

  gotoxy(1,4);

}

void refac(void)

{

  getch();

  clrscr();

}

Programul ASM2 compară două şiruri de caractere. Se citesc şirurile apelând întreruperea 21h cu funcţia 01h. Pentru comparare se apelează funcţia stringcmp() scrisă în limbajul C. Rezultatul returnat este interpretat în programul scris în limbaj de asamblare.

;ASM 2

.model small

.stack 100h

.data

  sir db 'parola',0

  sirc db 10 dup(?)

  m1 db 10,'Siruri identice',0

  m2 db 10,'Siruri diferite',0

  m3 db 10,'Dati sirul:',0

.code

  public _main

  extrn c printf:proc

  extrn _stringcmp:proc

  _main proc

    lea ax,m3

    call printf c,ax

    xor bx,bx

    citire:

      mov ah,01h

      int 21h

      cmp al,0dh

      je afisare

      mov sirc[bx],al

      inc bx

      jmp citire

    afisare:

      mov al,0

      mov sirc[bx],al

      lea ax,sir

      lea bx,sirc

      push ax

      push bx

      call _stringcmp

      add sp,4

      test ax,ax

      jnz diferite

      lea ax,m1

      call printf c,ax

      jmp sfarsit

      diferite:


lea ax,m2


call printf c,ax

    sfarsit:

      ret

  _main endp

end

//C2

int stringcmp(char* x,char* y)

{

  while(*x==*y)

  {

    if(*x=='\0')

      return 0;

    x++;

    y++;

  }

  return *x-*y;

}

Programul ASM3 concatenează două şiruri de caractere. Se citesc şirurile şi se apelează funcţia stringcat() care primeşte pe stivă adresele celor două variabile, sir1, sir2. Adresa şirului rezultat este returnată în registrul AX (funcţia returnează pointer spre char).

;ASM 3

.model small

.stack 100h

.data

  sir1 db 'mergem la ',0,20 dup(?)

  sir2 db 'stadion',0

  format db 'Sirul concatenat este : %s',10,13,0

.code

  public _main

  extrn c printf:proc

  extrn _stringcat:proc

  _main proc

    lea ax,sir1

    lea bx,sir2

    push bx

    push ax

    call _stringcat

    add sp,4

    lea bx,format

    call printf c,bx,ax

    ret

  _main endp

end

//C3

#include <conio.h>

char* stringcat(char* s1,const char* s2)

{

  char* t=s1;

  while(*s1)

    s1++;

  while(*s1++=*s2++);

  return t;

}

char* stringcat(char* s1,char* s2)

{

  int i;

  for(i=0;s1[i];i++);

  for(;s1[i++]=*s2++;);

  return s1;

}

Programul ASM4 afişează minimul sau maximul dintre două numere. Se transmit funcţiei scrise în limbajul C numerele şi opţiunea de calcul a valorii minime (1) sau maxime (2).

;ASM 4

.model small

.stack 100h

.data

  format1 db 'Introduceti a,b,optiune(1-min,2-max):',0

  format2 db '%d%d%d',0

  format3 db 'Functia este: %d',10,13,0

  a dw ?

  b dw ?

  x dw ?

  pf dw ?

.code

  public _main

  extrn _funct:proc

  extrn c printf:proc

  extrn c scanf:proc

  _main proc

    lea ax,format1

    call printf c,ax

    lea ax,format2

    lea bx,a

    lea cx,b

    lea dx,x

    call scanf c,ax,bx,cx,dx

    mov ax,x

    push ax

    mov dx,1

    call _funct

    add sp,2

    mov pf,ax

    call pf c,a,b

    lea bx,format3

    call printf c,bx,ax

    ret

  _main endp

end

//C4

int min(int a,int b)

{

  return a<b?a:b;

}

int max(int a,int b)

{

  return a>b?a:b;

}

int (*funct(int x))()

{

  int (*pfunct)();

  switch(x)

  {

    case 1:

      pfunct=min;

      break;

    case 2:

      pfunct=max;

      break;

  }

  return pfunct;

}
Programul ASM5 stabileşte şi afişează maximul dintre trei numere.

;ASM 5

.model small

mscanf macro par1,par2

  lea ax,par1

  lea bx,par2

  call scanf c,ax,bx

endm

mprintf macro par

  lea ax,par

  call printf c,ax

endm

.stack 100h

.data

  format1 db "Maxim = %d",10,13,0

  format2 db "a=",0

  format3 db "b=",0

  format4 db "c=",0

  format5 db "%d",0

  a dw ?

  b dw ?

  c dw ?

  rez dw ?

.code

  public _main

  extrn _max:proc

  extrn c scanf:proc

  extrn c printf:proc

  _main proc

    mprintf format2

    mscanf format5,a

    mprintf format3

    mscanf format5,b

    mprintf format4

    mscanf format5,c

    push c

    push b

    push a

    call _max

    add sp,6

    lea bx,format1

    call printf c,bx,ax

    ret

  _main endp

end

//C5

int max(int a,int b,int c)

{

  return a>b?(a>c?a:c):(b>c?b:c);

}

17.2 Programul principal este scris în limbajul C, procedurile apelate sunt scrise în limbaj de asamblare

Programul C6 citeşte un vector de numere întregi şi îl afişează pagină cu pagină. Afişarea se face cu ajutorul funcţiei afis() definită în limbaj de asamblare; aceasta apelează la rândul ei procedurile pozit() şi sterg().
//C6

#include <stdio.h>

#include <conio.h>

extern void afis(void);

void main(void)

{

  float v,x[40];

  int i,n;

  printf("n=");

  scanf("%d",&n);

  for(i=0;i<n;i++)

  {

    printf("x[%d]=",i);

    scanf("%f",&v);

    x[i]=v;

  }

  clrscr();

  for(i=0;i<n;i++)

    if((i%23==0)&&(i!=0))

    {

      afis();

      printf("\nx[%d]=%f]",i,x[i]);

    }

    else

      printf("\nx[%d]=%8.2f",i,x[i]);

  afis();

}

;ASM 6

; Afisarea unui mesaj, asteptarea apasarii unei taste

; si stergerea ecranului

.model small

.data

  sir db 'Pentru a continua apasati orice tasta...','$'

.code

  public _afis          
; Interfata cu programul C

  pozit proc            
; Pozitionarea pe linia 24, coloana 1

   mov ah,02h ; Functia BIOS

   mov dh,24  ; Linia

   mov dl,1   ; Coloana

   int 10h
  ; Apelul intr BIOS 10h (functii grafice)

   ret

  pozit endp

  sterg proc            ; Stergerea ecranului

    mov ah,0

    mov al,2

    int 10h

    ret

  sterg endp

  _afis proc

    call pozit          ; Pozitionare pe ecran

    mov ax,seg sir      ; Afisare mesaj

    mov ds,ax

    mov dx,offset sir

    mov ah,09h

    int 21h             ; Apel functie DOS afisare text terminat cu '$'

    mov ah,08h          ; Asteptarea apasarii unei taste

    int 21h

    call sterg

    ret                 ; Iesirea catre functia apelatoare

  _afis endp

end

Programul C7 calculează suma a trei numere citite de la tastatură. Pentru efectuarea adunării se apelează funcţia scrisă în limbaj de asamblare suma(). Parametrii sunt transmişi prin valoare, pe stivă.

//C7

#include <stdio.h>

extern int suma(int,int,int);

void main(void)

{

  int a,b,c;

  printf("a=");

  scanf("%d",&a);

  printf("b=");

  scanf("%d",&b);

  printf("c=");

  scanf("%d",&c);

  printf("suma=%d\n",suma(a,b,c));

}

;ASM 7

.model small

.code

  public _suma

  _suma proc

    push bp             ; Salvarea registrului bp

    mov bp,sp           ; bp<-sp

    mov ax,[bp+4]       ; ax=par1

    add ax,[bp+6]       ; ax+=par2

    add ax,[bp+8]       ; ax+=par3

    pop bp              ; Refacere bp

    ret                 ; Iesirea din functie

  _suma endp

end

Spre deosebire de programul precedent, programul C8 care efectuează aceleaşi calcule, transmite parametrii prin adresă, pe stivă.

//C8

#include <stdio.h>

extern int suma(int*,int*,int*);

void main(void)

{

  int *pa,a,*pb,b,*pc,c;

  printf("a=");scanf("%d",&a);

  printf("b=");scanf("%d",&b);

  printf("c=");scanf("%d",&c);

  pa=&a;pb=&b;pc=&c;

  printf("suma=%d\n",suma(pa,pb,pc));

}

;ASM 8

.model small

.stack 100h

.code

  public _suma

  _suma proc

    push bp             ; Salvare bp

    mov bp,sp           ; bp<-sp

    mov di,[bp+4]       ; di=pa

    mov ax,[di]         ; ax=*pa;

    mov di,[bp+6]       ; di=pb

    add ax,[di]         ; ax+=*pb

    mov di,[bp+8]       ; di=pc

    add ax,[di]         ; ax+=*pc

    pop bp              ; Refacere bp

    ret                 ; Iesire din functie

  _suma endp

end

Programul C9 exemplifică returnarea rezultatului pe stivă, prin adresă.

//C9

#include <stdio.h>

extern int* suma(int*,int*,int*);

void main(void)

{

  int *pa,a,*pb,b,*pc,c,*ps;

  printf("a=");

  scanf("%d",&a);

  printf("b=");

  scanf("%d",&b);

  printf("c=");

  scanf("%d",&c);

  pa=&a;

  pb=&b;

  pc=&c;

  ps=suma(pa,pb,pc);

  printf("suma=%d\n",*ps);

}

;ASM 9

.model small

.stack 100h

.data

  rez dw ?

.code

  public _suma

  _suma proc

    push bp

    mov bp,sp

    mov di,[bp+4]

    mov ax,[di]

    mov di,[bp+6]

    add ax,[di]

    mov di,[bp+8]

    add ax,[di]         ; ax=*par1+*par2+*par3

    mov rez,ax          ; rez<-ax

    mov ax,offset rez   ; ax<-&rez

    pop bp

    ret

  _suma endp

end

Programul C10 calculează şi afişează suma elementelor unui vector. Parametrii transmişi pe stivă sunt numărul de elemente al vectorului şi adresa primului element.

//C10

#include <stdio.h>

extern int suma(int*,int);

void main(void)

{

  int v,x[40];

  int i,n;

  printf("n=");

  scanf("%d",&n);

  for(i=0;i<n;i++)

  {

    printf("x[%d]=",i);

    scanf("%d",&v);

    x[i]=v;

  }

  printf("Suma elementelor vectorului = %d",suma(x,n));

}

;ASM 10

.model small

.stack 100h

.code

  public _suma

  _suma proc

    push bp

    mov bp,sp

    mov cx,[bp+6]       ; cx <- numarul de elemente

    mov si,[bp+4]       ; si <- adresa primului element

    xor ax,ax           ; ax=0

    ciclu:              ; do{

      add ax,[si]       ; ax+=x[i];

      inc si            ; i++

      inc si

      loop ciclu        ; }while(i<n)

    pop bp

    ret

  _suma endp

end

//C11

/*

  Produsul scalar a doi vectori.

*/

#include <stdio.h>

extern int prodscalar(int*,int*,int);

void main(void)

{

  int x[40],y[40],v,i,n;

  printf("n=");

  scanf("%d",&n);

  for(i=0;i<n;i++)

  {

    printf("x[%d]=",i);

    scanf("%d",&v);

    x[i]=v;

    printf("y[%d]=",i);

    scanf("%d",&v);

    y[i]=v;

  }

  v=prodscalar(x,y,n);

  printf("Produsul scalar = %d\n",v);

}

;ASM 11

; Produsul scalar a doi vectori

.model small

.stack 100h

.code

  public _prodscalar

  _prodscalar proc

    push bp

    mov bp,sp

    mov cx,[bp+8]
; Numarul de elemente

    mov si,[bp+4]
; Adresa primului vector

    mov di,[bp+6]
; Adresa celui de-al doilea vector

    xor bx,bx

    ciclu:

; Bucla repetata de cx ori

      mov ax,[si]
; bx+=[si]*[di];

      mul word ptr [di]

      add bx,ax

      inc si

      inc si

      inc di

      inc di

      loop ciclu

    mov ax,bx

; ax=bx (rezultatul)

    pop bp

    ret

  _prodscalar endp

end

//C12

/*

  Suma elementelor unei matrice.

*/

#include <stdio.h>

#define N 5

extern int suma(int x[][N],int m,int n);

void main(void)

{

  int x[3][N],v;

  int i,j,m,n;

  printf("numarul de linii = ");

  scanf("%d",&m);

  printf("numarul de coloane = ");

  scanf("%d",&n);

  for(i=0;i<m;i++)

    for(j=0;j<n;j++)

    {

      printf("x[%d][%d]=",i,j);

      scanf("%d",&v);

      x[i][j]=v;

    }

  printf("suma elementelor matricii = %d\n",suma(x,m,n));

}

;ASM 12

; Suma elementelor unei matrici

.model small

.stack 100h

.code

  c equ 5


; Numarul de coloane al matricii

  public _suma

  _suma proc

    push bp

    mov bp,sp

    mov bx,[bp+4]

; Adresa primului element

    xor ax,ax


; ax=0

    mov dx,[bp+6]

; Numarul de linii

    c1:

      mov cx,[bp+8]

; Numarul de coloane

      xor si,si


; si=0

      c2:


add ax,[bx][si]

; ax+=[bx][si]


inc si


; Urmatorul element


inc si


loop c2


; Mai sunt coloane ?

      add bx,c*2

; Urmatoarea linie

      dec dx

      jnz c1


; Mai sunt linii ?

    pop bp

    ret

  _suma endp

end

//C13

/*

Realizarea unei functii specificate de utilizator (min

sau max) asupra a doua numere introduse de la consola.

Se realizeaza cu aplelu unei functii asm care primeste

ca parametru

 functia dorita.

*/

#include <stdio.h>

extern int suma(int*,int*,int(*)(int,int),int);

int max(int a,int b)

{

  return a>b?a:b;

}

void main(void)

{

  int x[40],y[40],n,i,v;

  printf("n=");

  scanf("%d",&n);

  for(i=0;i<n;i++)

  {

    printf("x[%d]=",i);

    scanf("%d",&v);

    x[i]=v;

    printf("y[%d]=",i);

    scanf("%d",&v);

    y[i]=v;

  }

  printf("Suma elementelor maxime = %d\n",suma(x,y,max,n));

}

;ASM 13

; Suma elementelor maxime din doi vectori de elemente

; Functia de comparare e transmisa ca parametru

; (o adresa near)

.model small

.stack 100h

.data

  rez dw 0

.code

  public _suma

  _suma proc

    push bp

    mov bp,sp

    mov cx,[bp+10]

; Numarul de elemente

    mov di,[bp+6]

; Adresa celui de-al doilea vector

    mov si,[bp+4]

; Adresa primului vector

    ciclu:

      mov ax,[si]

      mov bx,[di]

      call word ptr [bp+8] c,ax,bx
; Apel functie de comparare

      add rez,ax

; rez+=max

      inc si


; Urmatoarele elemente din vectori

      inc si

      inc di

      inc di

      loop ciclu

    mov ax,rez


; ax<-rez

    pop bp

    ret

  _suma endp

end

Celelalte programe realizează prelucrări care evidenţiază modalităţi de dezvoltare software conform programării mixte.

În cazul în care se doreşte gestionarea strictă a ciclurilor maşină în secvenţe de program cu structuri repetitive imbricate în care numărul de referiri ale unei secvenţe este foarte mare, se preferă dezvoltarea unui program mixt în care secvenţa referită este scrisă direct în limbaj de asamblare. De exemplu, secvenţa:

...


s=0;


for (i=0; i<30000; i++)



for (j=0; j<70000; j++)




for (k=0; k<20000; k++)





if(i= =j)

s+=k*2;





else






s+=k/2;

...

se transformă în secvenţa mai performantă de mai jos:

...


s=0;


for (i=0; i<30000; i++)



for (j=0; j<70000; j++)




for (k=0; k<20000; k++)





{

asm mov bx, k

asm mov ax, i

asm cmp ax, j

asm jz egal

asm shr bx 

asm egal: nop

asm shl bx

asm mov ax, s

asm add ax, bx

asm mov s, ax





}

...



Blocul instrucţiunilor limbajului de asamblare se delimitează cu atributul asm, obţinându-se secvenţa echivalentă

asm
{

mov bx, k

mov ax, i

cmp ax, j

jz egal

shr bx 

egal: nop

shl bx

mov ax, s

add ax, bx

mov s, ax





}

17.3 Concluzii


Programarea mixtă are ca obiective atât creşterea performanţei programelor în execuţie, cât şi reutilizarea de software verificat în practică, din diferite generaţii, scris în limbaje de programare mai puţin utilizate în prezent.


De asemenea prin programarea mixtă se integrează programele scrise în orice limbaj de programare formând un produs software complex.

